

MODEL:

SETS:

months/1..6/:P,W,O,H,F,S,I,B,WD,D;

ENDSETS

!Función objetivo;

min=@sum(months(t):pc*P(t)+15*8*WD(t)*W(t)+oc*O(t)+hc*H(t)+fc*F(t)+sc*S(t)+ic*I(t)+bc*B(t));

! Productos fabricados + inventario anterior + productos por subcontratación + faltante del mes = demanda + faltante del mes anterior + inventario final;

@for(months(t) | t#GT#1:[Demanda_del_mes_de]

P(t)+I(t-1)+S(t)+B(t)-I(t)-B(t-1) = D(t););

! Productos en el mes 1 + inventario inicial + productos subcontratados + productos faltantes en el 1 = demanda en el mes + productos faltantes en el mes 0 + inventario final del mes 1;

[Demanda_del_mes_1_]P(1)+I0+S(1)+B(1)-I(1)-B0 = D(1);

! Trabajadores del mes - trabajadores del mes anterior = trabajadores contratados - trabajadores despedidos;

@for(months(t) | t#GT#1:[Balance_de_trabajadores]

W(t)-W(t-1)-H(t)+F(t) = 0;);

[Balance_de_trabajadores_1_]W(1)-W0-H(1)+F(1)=0;

! Horas de producción + horas extra<horas trabajables;

@for(months(t):[Capacidad_de_horas]

5*P(t)-8*WD(t)*W(t)-O(t) < 0;);

!Inventario al final del mes> 25%de la demanda del mes;

@for(months(t):[Inventario_minimo]

I(t) > 0.25*D(t););

! No deben faltar productos en último mes;

[Stockout_mes_6_]B(6) = 0;

DATA:

D=1800,1500,1100,900,1100,1600;!Demanda;

WD=22,19,21,21,22,20;!Días trabajables;

pc=100;!Costo de producción;

oc=20;!Costo de mano de obra;

hc=1000;!Costo de contratación;

fc=1500;!Costo de despidos;

ic=5;!Costo de inventario;

bc=10;!Costo de Stockout;

sc=120;!Costo de subcontratación;

I0=400;!Inventario inicial;

B0=0;!Stockout inicial;

W0=38;!Trabajadores iniciales;

ENDDATA

END